Stochastic Programming in a Complementarity Format: Tools and Sample Applications

Thomas F. Rutherford[1]

Center for Energy Policy and Economics
ETH Zürich

Talk Presented at the University of Ottawa

Wednesday, June 24, 2009

[1] This talk is based on joint work with Alex Meeraus. I also wish to acknowledge several helpful comments and suggestions from Olivier Bahn and Alain Haurie.

Programming Tools for Stochastic Programs

1. Event tree management logic tool (`probtree`)
2. Our approach accomodates *tight formulations* based on a minimal set of decision variables and constraints.
3. Graphical tools for debugging (`treeplot`) and reporting (`fanplot`)
4. General objective is to accomodate complementarity programming in a stochastic framework. Tools are useful in both optimization and equilibrium applications.

Two illustrative models

1. Lifecyle consumption-savings decisions with income uncertainty (finite horizon NLP)
2. Ramsey growth model with uncertain technology change (infinite-horizon NLP)

A Simple Lifecycle Model

- A lifecycle savings investment model in which there is income uncertainty maximizing the discounted expected utility
- Utility function: Logarithm of consumption
- Version 1: Borrowing and savings
- Version 2: Only savings

Earnings Profiles in Four States of Nature



Tree Plot of Lifecycle Event Tree



Transition Structure Specification in GAMS

```
sets
  t   Time periods for a typical life-cycle (ages) /20*75/

  sw  States of world / 30pa peak earn  30K per year
                        40pa peak earn  40K per year
                        80pa peak earn  80K per year
                       100pa peak earn 100K per year /

  transition(t,sw,sw)     State transitions /
   20.40pa. 30pa    Learn at age 20 if you are going to college
   25.40pa. 80pa    Learn at age 25 if you earn a PhD degree
   35.80pa.100pa    Learn at age 35 that you are good at business /

parameter pi(sw) Subjective probability /
                        30pa 0.3
                        40pa 0.4
                        80pa 0.2
                       100pa 0.1 /
```

Probability-Scaled Plot of Lifecycle Event Tree



Data Structures for Stochastic Programming

```
sets
  eq(t,sw)      Equilibrium structure: sw is active in t
  st(t,sw,sow) State transitions: sw transitions to sow in t
  sm(t,sw,sow) State matching: sw is represented by sow in t;

* Use probtree to load data structures:

$batinclude probtree t sw transition eq st sm

b.lo(t,sw)=-inf; b.up(t,sw)=+inf;       // Borrowing and saving permitted
b.fx(tlast,sw) = 0;                     // No debt or savings at the end

solve lcycle using nlp maximizing eu;   // Solve the stochastic (ATL) model

loop(sw, pi(sow)=0; pi(sw)=1;           // Solve deterministic (LTA) models

  solve lcycle using nlp maximizing eu; );
```

Consumption Profiles with Borrowing (ATL vs. LTA)



Consumption Profiles with Liquidity Constraints



Asset Balances with Borrowing



Asset Balances without Borrowing



Lessons from the Lifecycle Model

1. Event tree representation via state transitions using `probtree` utility
2. Event tree visualization using `treeplot` utility.
3. Without hedging possibilities stochasticity cannot be exploited
4. Importance of visual presentation of input and output

The Ramsey Model (implicit algebra)

$$\max E(\sum_{t=0}^{\infty} \beta_t u(\tilde{c}_t))$$

s.t.

$$\tilde{c}_t + \tilde{\phi}_t \tilde{i}_t = \tilde{y}_t$$
$$\tilde{y}_t = f(\tilde{k}_t)$$
$$\tilde{k}_{t+1} = (1-\delta)\tilde{k}_t + \tilde{i}_t$$
$$\tilde{i}_t \geq 0$$

Data Structures for Stochastic Programming

- s (`sw`) Set of scenarios, associated with all leaves at the bottom of the event tree.
- t (`t`) Set of time periods in the model. There must be at least as many time periods in the model as there are levels in the event tree.
- $\mathcal{S}_t$ (`eq`) Each node in the event tree is labelled with a scenario index s. Set $\mathcal{S}_t$ defines the set of "active scenarios" in time period t. If a scenario has appeared in time period t, it must be present in all subsequent time periods:

$$s \in \mathcal{S}_t \Rightarrow s \in \mathcal{S}_\tau \forall \tau > t$$

Data Structures (cont.)

$\mathcal{T}_{st}$ (`st`) Defines the scenarios which branch on from scenario s in time period t – the *transition points*. Once a scenario has “appeared”, it must follow itself in all subsequent periods, i.e.:

$$s \in \mathcal{S}_t \Rightarrow s \in \mathcal{T}_{s\tau} \forall \tau \geq t$$

Ω_{ts} (`sm`) Defines the set of scenarios which branch from scenario s or from a “descendent of s” in time period t or in any later period:

$$s' \in \Omega_{st} \Rightarrow \exists t' \geq t \text{ such that } s' \in \mathcal{T}_{st'}$$

Ramsey Model (explicit algebra)

$$\max \sum_s \pi_s \left(\sum_{t=0}^{\infty} \beta_t u(c_{st}) \right)$$

s.t.

$$c_{st} + i_{st} = y_{st}$$

$$y_{st} = f(k_{st})$$

$$k_{s',t+1} = (1-\delta)k_{st} + \phi_{st} i_{st} \quad \forall s' \in \mathcal{T}_{st}$$

and *non-anticipativity constraints*:

$$\left.\begin{aligned} c_{s'\tau} &= c_{st} \\ k_{s'\tau} &= k_{st} \\ y_{s'\tau} &= y_{st} \\ i_{s'\tau} &= i_{st} \end{aligned}\right\} \forall s' \in \Omega_{st}, \tau \le t$$

Ramsey Model (tight formulation)

$$\max \sum_s \pi_s \left(\sum_{t=0}^{\infty} \sum_{(s' | s \in \Omega_{ts'})} \beta_t u(c_{s't}) \right)$$

s.t.

$$\begin{aligned} c_{st} + i_{st} &= y_{st} && \forall s \in \mathcal{S}_t \\ y_{st} &= f(k_{st}) && \forall s \in \mathcal{S}_t \\ k_{s',t+1} &= (1-\delta)k_{st} + \phi_{st} i_{st} && \forall s' \in \mathcal{T}_{st}, \forall s \in \mathcal{S}_t \end{aligned}$$

N.B.

- *Non-anticipativity constraints* are unnecessary.
- $\phi_{st} \geq 1$ describes the productivity of investments undertaken in state s, time period t.

Recovery of the Solution from a Tight Formulation

$$\left.\begin{array}{l} c_{s'\tau} \leftarrow c_{st} \\ k_{s'\tau} \leftarrow k_{st} \\ y_{s'\tau} \leftarrow y_{st} \\ i_{s'\tau} \leftarrow i_{st} \end{array}\right\} \forall s' \in \Omega_{st}, \tau \leq t$$

Probability of Efficiency Improvement



Cumulative Probability Density Function



GAMS Code: Building the Event Tree

```
set t                              Time periods in the model /2005*2100/
    sw                             States of world /2010*2030,never/
    transition(t,sw,sw)            State transitions;

* Define the event tree by specifying which states
* generate transitions. In this case 2010 defines the root
* node of the tree.

transition("2010","2010","2011") = yes;
loop((t,sw)$transition(t,sw,sw+1),
  transition(t+1,sw+1,sw+2) = yes);

*       Apply a normal distribution over the different states:

pi(sw) = exp(-var*sqr(ord(sw)-card(sw)/2));

*       Normalize:

pi("never") = 0; pi(sw) = 0.75*pi(sw)/sum(sow,pi(sow));
pi("never") = 1 - sum(sw,pi(sw));
```

Ramsey Event Tree



Ramsey Event Tree (`gams treeplot --piscale`)



State Contingent Investment



Investment Response: The Fan Plot Tool (Decile Shading)



Lessons Learned

1. We present two mathematically equivalent formulations: with and without non-anticipativity constraints (NAC). The *tight formulation* is preferred (no NAC) with easy recovery of solution of the explicit problem.
2. Construction of transition matrix
3. Discretization of continuous distributions
4. Use of FanPlot for visualization of model output.

Limitations of Stochastic Programming

1. Probability of events are *controlled by nature*.
2. The actions of agents within the model are unable to alter the probability of particular outcomes.
3. It is difficult to use this framework to study issues related to R&D investments.

Climate R&D Policy: Key Assumptions

1. R&D outcomes are uncertain.
2. R&D is costly and shared inputs are scarce.
3. The probability of success in R&D increases with expenditures.
4. R&D is a portfolio problem, and optimal policy should be conditioned on available information.

Model Inputs

- j Set of technologies
  - CCS Carbon capture and storage,
  - GEN4 Advanced nuclear power,
  - SOLAR Solar power
- mc_j Marginal cost if successful
- K_j Output capacity if successful
- q_j Ex-ante probability of marketability
- $\bar{z}_j$ Mean research and development cost
- $F_j(z)$ Cumulative density function: probability of a breakthrough for technology j after a research expenditure z.

Technology Data

	R&D Cost ($\bar{z}_j$)	Mrg. Cost (mc_j)	Capacity (K_j)	Probability (q_j)
CCS	75	200	100	0.4
GEN4	50	100	100	0.2
SOLAR	25	400	200	0.6

Stochastic Structure

s States of world

- REF Market value of carbon increases from \$50 per ton in 2010 to \$550 per ton in 2050
- HI Market value of carbon increases from \$50 per ton in 2010 to \$2050 per ton in 2050
- LO Market value of carbon declines at some point to \$50 per ton in 2050

t_s Year in which scenario s diverges from reference

States of World and Stochastic Structure



Future Value Carbon Prices



Decision Variables

For each technology j, time period t and state of world s we keep track of the following non-negative values:

$x_{j,t,s}$ R&D investment during the period,

$p_{j,t,s}$ Probability that no discovery has been made,

$q_{j,t,s}$ Ex-ante probability that breakthrough is possible,

$z_{j,t,s}$ Cumulative R&D at the start of the period

While all variables enter simultaneously into the nonlinear program, it is convenient to think of $x_{j,t,s}$ as the control variable and all of the others as state variables.

Technology Combinations

Set *k* consists of the following elements:

- Singleton technologies (CCS, GEN4, SOLAR)
- Pairs of technologies (CCS_GEN4, CCS_SOLAR, GEN4_SOLAR)
- All three technologies (CCS_GEN4_SOLAR)

We subsequently use the Kroniker delta function to indicate which technologies are in which sets:

$$\delta_{jk} = \begin{cases} 1 & \text{if} \quad \mathrm{j} \in \mathrm{k} \\ 0 & \text{if} \quad \mathrm{j} \notin \mathrm{k} \end{cases}$$

Intermediate Variables: Computed Probabilities

$\omega_{k,t,s}$ Probability that only technologies in combination k have been developed:

$$\omega_{k,t,s} = \prod_j \left[\delta_{jk} \times (1 - P_{j,t,s}) + (1 - \delta_{jk}) \times P_{j,t,s}\right]$$

$\gamma_{k,t,s}$ Probability that only the technologies in combination k have been undeveloped:

$$\gamma_{k,t,s} = \prod_j \left[\delta_{jk} \times P_{j,t,s} + (1 - \delta_{jk}) \times (1 - P_{j,t,s})\right]$$

Probability of a Breakthrough

N.B. In the following, we suppress subscripts j, t, s where possible to simplify notation.

The probability of a breakthrough in the current period is conditioned on cumulative investment to date, z, and a Bayesian prior probability that the technology is ultimately marketable, q.

Then, the probability of a breakthrough following an additional investment of x is given by:

$$PB(x|q,z) = q \times \frac{F(z+x) - F(z)}{1 - F(z)}$$

Alternative Probability Distributions

1. Exponential

$$F_j(z) = 1 - e^{\lambda_j z}$$

where $\lambda_j = 1/\bar{z}_j$

2. Raleigh

$$F_j(z) = 1 - e^{-(z/\sigma_j)^2/2}$$

where

$$\sigma_j = \bar{z}_j/(\pi/2)$$

Bayesian Learning

If a research expenditure is made in the current period and no breakthrough is observed, the research manager then reassesses the technology's potential. We model this updating using Bayesian inference:

$$P(A|B) = P(A)\,\frac{P(B|A)}{P(B)}$$

in which we interpret events A and B as

- A the technology is potentially successful
- B the technology fails to produce a breakthrough in current period

Bayesian Update

Applying this idea to the current model, we interpret $P(A|B)$ as Q_{t+1}, and we interpret $P(A)$ as the current subjective probability, Q_t.

We have a failed R&D investment of x in period t given current cumulative investment z. We thus have:

$$P(B|A) = \frac{1 - F(z + x)}{1 - F(z)}$$

and

$$P(B) = 1 - q_t \times \frac{F(z + x) - F(z)}{1 - F(z)}.$$

Hence,

$$q_{t+1} = q_t \times \frac{1 - F(z + x)}{1 - F(z) - q_t \times (F(z + x) - F(z))}.$$

Accounting Conditions and Growth Constraints

1. Probability of no breakthrough in time period $t + 1$ is equal to the probability that there had been no breakthrough prior to year t and none occurred during period t:

$$P_{t+1} = P_t \times (1 - PB_t)$$

2. Cumulative research and development expenditures:

$$z_{t+1} = z_t + x_t$$

3. Expansion and decline constraint:

$$(1 - \Delta)x_t \leq x_{t+1} \leq (1 + \chi)x_t + \xi\bar{z}$$

in which Δ, χ and ξ are maximum rates of decline, expansion and initial period expenditure fraction.

Costs and Benefits

Define:

pv_t Present value price index to time period t:

$$pv_t = \left(\frac{1}{1+r}\right)^t$$

π_s Probability of state of world s

pc_{ts} Future value carbon price in time period t and state of world s

The Maximand

The expected present value of profits associated with the research and development plan $x_{j,t,s}$ is given by:

$$EPV = \sum_{t,s,k} pv_t \pi_s \left(\omega_{k,t,s} R_{k,t,s} - \gamma_{k,t,s} C_{k,t,s} \right)$$

where $R_{k,t,s}$ is revenue which would result if technologies in collection k are available:

$$R_{k,t,s} = \sum_{j \in k} \max(0, pc_{ts} - mc_j) \times K_j,$$

and $C_{k,t,s}$ is the aggregate cost of research and development expenditure in time period t, state of world s when only technologies in set k have been discovered. This is assumed to increase with the fourth power of the aggregate expenditure, representing in a stylized way the idea that R&D compete for scarce resources:

$$C_{k,t,s} = \left(\sum_{j \notin k} x_{j,t,s} \right)^4$$

GAMS Code – Sets

```
*       Define the default stochastic structure:

$if not set swfile $set swfile atl

set     t       Time periods /2010*2150/,
        t0(t)   Base year /2010/,

        j       Technologies /
                        ccs     Carbon capture and storage,
                        gen4    Advanced (generation 4) nuclear power,
                        solar   Solar power /,

        k       Technology outcomes /
                ccs, gen4, solar,
                ccs_gen4, ccs_solar, gen4_solar,
                ccs_gen4_solar /,

        kj(k,j)  Match of technology to scenario
                /ccs.ccs,gen4.gen4,solar.solar,
                ccs_gen4.(ccs,gen4),ccs_solar.(ccs,solar),gen4_solar.(gen4,solar),
                ccs_gen4_solar.(ccs,gen4,solar)/;

alias (j,j_)

*       Read the state of world file which describes future price
*       scenarios:

$include %swfile%.gms
```

GAMS Code – Data

```
table tech(j,*) Technology description
                rdcost  mc      capacity        q0
        ccs       75    200     100             0.4
        gen4      50    100     100             0.2
        solar     25    400     200             0.6;

parameter
        rdcost(j)       Mean investment required to achieve breakthrough
        mc(j)           Marginal cost of supply
        capacity(j)     Upper bound on market supply
        q0(j)           Ex-ante probability of success;

q0(j)       = tech(j,"q0");
rdcost(j)   = tech(j,"rdcost");
capacity(j) = tech(j,"capacity");
mc(j) = tech(j,"mc");

parameters
        ir              Interest rate /0.05/
        xlim            Introduction bound on X (fraction of rdcost) /0.02/,
        expf            Expansion factor (per annum) /0.1/,
        decf            Decline factor   (per annum) /0.2/,
        Exponential     Flag for the Exponential distribution /0/,
        Raleigh         Flag for the Raleigh distribution /1/,
        lamda           Exponential distribution parameter
        sigma           Raleigh distribution parameter
        pv(t)           Present value price index,
        mv(j,t,sw)      Marginal value of future supply;

pv(t)       = (1+ir)**(1-ord(t));
mv(j,t,sw)  = max(0,fvprice(t,sw)-mc(j));
lamda(j)  = 1/rdcost(j);
sigma(j)  = rdcost(j)/sqrt(3.141592/2);
```

GAMS Code – Event Tree for “Act then Learn”

```
$title  Act-Then-Learn Model

set     sw      States of world /hi2015,hi2020,hi2025,hi2030,ref,lo2030/

        transition(t,sw,sw)     State transitions
                / 2015.ref.hi2015,  2020.ref.hi2020,
                  2025.ref.hi2025,  2030.ref.hi2030,  2030.ref.lo2030 /;

parameter       pi(sw)  Probability of climate emissions constraint /
        ref 0.60, hi2015 0.05, hi2020 0.05, hi2025 0.05, hi2030 0.05, lo2030 0.2/;

execute_unload 'probtree.gdx', t, sw, transition;
execute 'gams probtree gdx=treelogic.gdx';

set     eq(t,sw)        Equilibrium structure (sw is active in time period tp),
        st(t,sw,sow)    State transitions (sw transitions to sow in time period t),
        sm(t,sw,sow)    State matching (sw is represented by state sow in time period t);

execute_load 'treelogic.gdx', eq, st, sm;

parameter       ctax(sw)        /ref 500, lo2030 0, (hi2015,hi2020,hi2025,hi2030) 2000/,
                fvprice(t,sw)   Future value carbon tax rate;

fvprice(t,sw) = 50 + (ord(t)-1)/(card(t)-1) * sum(sm(t,sw,sow), ctax(sow));
```

GAMS Code – Event Tree for “Learn then Act”

```
$if not set ttax $set ttax 2030
$if not set ctax $set ctax 2000
$if %sw%==ref    $set ttax 1e6
$if %sw%==ref    $set ttax 2010
$if %sw%==hi2015 $set ttax 2016
$if %sw%==hi2020 $set ttax 2021
$if %sw%==hi2025 $set ttax 2026
$if %sw%==hi2030 $set ttax 2031
$if %sw%==lo2030 $set ttax 2031
$if %sw%==lo2030 $set ctax 0

set     sw                        States of world /%sw%/,
        transition(t,sw,sw)       State transitions
        eq(t,sw)                  Equilibrium structure
        st(t,sw,sw)               State transitions
        sm(t,sw,sw)               State matching;

parameter       ctax(sw)          Carbon tax rate,
                pi(sw)            Probability of climate emissions constraint,
                fvprice(t,*)      Future value carbon tax rate;
ctax(sw) = %ctax%;
transition(t,sw,sw) = yes;
eq(t,sw) = yes;
st(t,sw,sow) = yes;
sm(t,sw,sow) = yes;
set tl(t)/2020,2040,2060,2080,2100/;
pi(sw) = 1;
fvprice(t,"ref") = 50 + (ord(t)-1)/(card(t)-1) * 500;
fvprice(t,sw) = 50 + (ord(t)-1)/(card(t)-1) * 500;
fvprice(t,sw)$(t.val >= %ttax%) = 50 + (ord(t)-1)/(card(t)-1) * ctax(sw);
```

GAMS Code – Macro Definitions

```
$macro CDFexp(j,z)  (1 - exp(-lamda(j)*z))

$macro CDFral(j,z)  (1-exp(-z*z/(2*sigma(j)*sigma(j))))

$macro F(j,z) (CDFexp(j,z)$Exponential + CDFral(j,z)$Raleigh)

$macro OMEGA(k,t,sw) prod(j_,(1-P(j_,t,sow))$kj(k,j_)+P(j_,t,sow)$(not kj(k,j_)))

$macro GAMMA(k,t,sw) prod(j_,(1-P(j_,t,sow))$(not kj(k,j_))+P(j_,t,sow)$kj(k,j_))

$macro PB(j,t,sw) (Q(j,t,sw)*(F(j,(Z(j,t,sw)+X(j,t,sw)))-CDF(j,t,sw))/(1-CDF(j,t,sw)))
```

GAMS Code – Variables and Equations

```
positive
variables
        P(j,t,sw)       Probability that no discovery has been made by t,
        Q(j,t,sw)       Ex-ante probability that breakthrough is possible,
        X(j,t,sw)       R&D investment at during the period,
        Z(j,t,sw)       Cumulative R&D at the start of the period,
        CDF(j,t,sw)     Cumulative density;

variable        OBJ     Objective function (present value);

equations       objdef, pdef, zdef, qdef, cdfdef, decline, expand;
```

GAMS Code – Equation Definitions

```
objdef..        OBJ =e= sum(sm(t,sw,sow), pv(t) * pi(sw) * sum(k,

*       Include all combinations of marketed technologies.  The
*       consumer surplus should account for both the abatement
*       potential and the carbon price for each possibility
*       configuration of breakgthroughs:

*       Profits for existing technologies:

                OMEGA(k,t,sow) * sum(kj(k,j), mv(j,t,sow)*capacity(j))

*       Account for the cost of all possible combinations of
*       technologies which are under development.  We assume that
*       there are pecuniary externalities and diminishing returns to
*       scale in the level of aggregate R&D.  The aggregate level
*       of R&D cost increases with scale:

              - GAMMA(k,t,sow) * power(sum(j$(not kj(k,j)),X(j,t,sow)),4)));
```

GAMS Code – Equation Definitions (cont.)

```
*       Probability that a technology remains undiscovered in the next
*       period depends on the probability that it was undiscovered at
*       the start of the previous period, and no breaktrhough occurred
*       in the previous period:

pdef(j,t+1,sw,sow)$st(t,sw,sow)..

        P(j,t+1,sow) =e= P(j,t,sw) * (1-PB(j,t,sw));

*       We define the cumulative density function as a variable in order to place
*       an upper bound just below unity on its value:

cdfdef(j,eq(t,sw))..

        CDF(j,t,sw) =e= F(j,Z(j,t,sw));

*       R&D investment in one period is added to the cumulative
*       investment stock at the outset of the subsequent period:

zdef(j,t+1,sw,sow)$st(t,sw,sow)..

        Z(j,t+1,sow) =e= Z(j,t,sw) + X(j,t,sw);
```

GAMS Code – Equation Definitions (cont.)

```
*       Bayesian inference updates the probability of ultimate
*       success in the subsequent period following failure:

qdef(j,t+1,sw,sow)$st(t,sw,sow)..

        Q(j,t+1,sow) =e= Q(j,t,sw) * (1-F(j,(Z(j,t,sw)+X(j,t,sw)))) /

                (1 - CDF(j,t,sw) - Q(j,t,sw)*(F(j,(Z(j,t,sw)+X(j,t,sw)))-CDF(j,t,sw)));

*       We include some ad-hoc constraints which prevent bang-bang
*       outcomes.  We will ultimately need to think about whether
*       these are sufficiently unattractive that we would want to
*       instead include R&D capital stocks which accomplish the same
*       thing:

decline(j,t+1,sw,sow)$st(t,sw,sow)..

        X(j,t+1,sow) =G= (1-decf) * X(j,t,sw);

expand(j,st(t+1,sw,sow))..

        X(j,t+1,sow) =L= (1+expf) * X(j,t,sw) + rdcost(j)*xlim;
```

GAMS Code – Initialization and Solution

```
model optrd /all/;

P.L(j,eq(t,sw)) = 1; Z.L(j,eq(t,sw)) = 1; Q.L(j,eq(t,sw)) = 1; X.L(j,eq(t,sw)) = 1;

P.FX(j,eq(t0,sw)) = 1;
Q.FX(j,eq(t0,sw)) = q0(j);
Z.FX(j,eq(t0,sw)) = 0;
X.UP(j,eq(t0,sw)) = xlim*rdcost(j);

CDF.UP(j,eq(t,sw)) = 0.999;
CDF.FX(j,t,sw)$(not eq(t,sw)) = 0;

Exponential=1; Raleigh=0;
solve optrd using nlp maximizing obj;

$ondotl
parameter pivotdata     Pivot Table Database;
loop(sm(t,sow,sw),
  pivotdata("p",t,j,sow)  = P.L(j,t,sw);
  pivotdata("x",t,j,sow)  = X.L(j,t,sw);
  pivotdata("z",t,j,sow)  = Z.L(j,t,sw);
  pivotdata("q",t,j,sow)  = Q.L(j,t,sw);
  pivotdata("pb",t,j,sow) = PB(j,t,sw);
);
```

Computational Experience

Computational cost for LTA on a Lenovo X61 portable computer with an Intel T7500 processor @ 2.20 GHz:

```
MODEL STATISTICS

BLOCKS OF EQUATIONS           7     SINGLE EQUATIONS       13,861
BLOCKS OF VARIABLES           6     SINGLE VARIABLES       11,581  1 projected
NON ZERO ELEMENTS        50,827     NON LINEAR N-Z         27,729
DERIVATIVE POOL           4,637     CONSTANT POOL           1,378
CODE LENGTH             540,267

...

 RESOURCE USAGE, LIMIT        214.140      1000.000
 ITERATION COUNT, LIMIT      3601         10000
 EVALUATION ERRORS              0             0
```

This is an extremely low computational cost, *essentially free*.

How did the rabbit get into the hat?

The key simplification in this model formulation centers on the role of the key control variables, x_{jst}. The set describing alternative technology combinations, k, enters into the definition of the objective function, but it does not appear as an argument in any of the variables or constraints.

If we were to condition decisions on k, problem dimensionality would increase geometrically, and simultaneous solution would most likely would be impossible.

Breakthrough Probabilities: ATL vs. LTA



Breakthrough Probabilities: CCS (ATL vs. LTA)



Breakthrough Probabilities: Solar (ATL vs. LTA)



Breakthrough Probabilities: Gen4 (ATL vs. LTA)



R&D Expenditures: CCS (ATL vs. LTA)



R&D Expenditures: Solar (ATL vs. LTA)



R&D Expenditures: Gen4 (ATL vs. LTA)

Conclusions

1. *Simultaneous solution* of certain classes of stochastic control problems is perhaps more useful for policy-relevant analysis than has been previously assumed.
2. Empirical assessment of this model is an area of active research. What robust hedging strategies can be identified on the basis of current perceptions about the costs and benefits of alternative R&D policy proposals?
3. This framework is limited to *finite state* / *infinite horizon* models. The resulting models can be used to study transition policies, contingent on long-term deterministic steady-state growth paths.